

(* Asymptotic expected number of hairpins in ALL structures with theta=1 and stickiness p=1*)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota–Lalley–Woods Theorem. Grammar with S, T where T means there is at least one base pair. *)

```

Clear["*"]
(* S arrow D or T; T arrow (E) or T* or (T) or S(D1) or S(T); D0 arrow * or D0 *;
D1 arrow *** or D1 * *)
Clear[p, S, D0, D1, z, eqn0, eqn, F, z0, y0, dFdZOfz0S0, d2FdYOfz0S0];

eqn0 = {S == z + z S + z^2 S + z^2 S^2}; (* grammar for all str, theta=1,p=1 *)
CellPrint["eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1"]
eqn1 = {S == D0 + T, T == z^2 D1 + z T + z^2 T + z^2 S T + z^2 S D1,
        D0 == z + z D0, D1 == z + z D1};
CellPrint[
  "eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0"]
Eliminate[eqn0, {D0, D1, T}]
Collect[%, S, Simplify]
Eliminate[eqn1, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["We derive the SAME equations
  from eqn0,eqn1 showing the underlying grammar is correct."]
p = 1;
eqn = {S == D0 + T, T == p z^2 D1 + z T + p z^2 T + p z^2 S T + p z^2 S D1,
        D0 == z + z D0, D1 == z + z D1};
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S = ... to get F"]
F = (S^2 z^2 + z) / -(-1 + z + z^2);

NSolve[{F == S, D[F, S] == 1}, {z, S}]
z0 = 0.3819660112501051`;
y0 = 1.6180339887498951`;

dFdZOfz0S0 = D[F, z] /. {z -> z0, S -> y0}

d2FdYOfz0S0 = D[F, {S, 2}] /. {z -> z0, S -> y0}
c = Sqrt[z0 dFdZOfz0S0 / (2 Pi d2FdYOfz0S0)]
c * (1 / z0) ^ n n ^ (-3 / 2)

```

eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1

eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

$$S^2 z^2 + S(-1 + z + z^2) = -z$$

$$S^2 z^2 + S(-1 + z + z^2) = -z$$

We derive the SAME equations from eqn0,eqn1 showing the underlying grammar is correct.

$$S^2 z^2 + S(-1 + z + z^2) = -z$$

$$S^2 z^2 + S(-1 + z + z^2) = -z$$

From previous expression, write $S = \dots$ to get F

$$\left\{ \left\{ z \rightarrow 2.61803, S \rightarrow -0.618034 \right\}, \left\{ z \rightarrow 0.381966, S \rightarrow 1.61803 \right\}, \right. \\ \left. \left\{ z \rightarrow -0.5 + 0.866025 i, S \rightarrow -0.5 + 0.866025 i \right\}, \left\{ z \rightarrow -0.5 - 0.866025 i, S \rightarrow -0.5 - 0.866025 i \right\} \right\}$$

$$12.3992$$

$$0.618034$$

$$1.10437$$

$$\frac{1.10437 \times 2.61803^n}{n^{3/2}}$$

(*Now, we compute mean, variance using Drmota's Theorem*)

```

Clear["*"]
p = 1;
eqn = {S == D0 + T, T == p u z^2 D1 + z T + p z^2 T + p z^2 S T + p u z^2 S D1,
      D0 == z + z D0, D1 == z + z D1};

CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S=... to obtain F"]
F = (S^2 (-1 + z) z^2 + z (-1 + z + z^2) - (u z^3 + S u z^3)) / - (1 - 2 z + 2 z^3);

f = (F /. S -> s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]
c = Denominator[Together[s - f]]
dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
d = Denominator[Together[1 - dfs]]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
If[Denominator[Together[s - f]] == Denominator[Together[1 - dfs]],
  CellPrint["Denominator of (s-f) same as that of (1-dfs)"],
  CellPrint["Denominator of (s-f) different than than of (1-dfs)"]]
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]

(* res =Resultant[s-f,1-dfs,s] *)

```

```

(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)

res /. z -> z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)

rho = 0.3819660112501051`;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u -> 1) /. z[1] -> rho
CellPrint[
  "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u -> 1) /. z[1] -> rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = (((D[dzu, u] /. u -> 1) /. z[1] -> rho) /. z'[1] -> dzuEvaluatedAt1)
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu
CellPrint["Now compute the standard deviation sigma"]
stdev = Sqrt[var]

```

Eliminate all variables except S,u,z

$$S - z - 2 S z + z^2 - S^2 z^2 + z^3 + 2 S z^3 + S^2 z^3 = (1 + S) u z^3$$

$$S^2 (-1 + z) z^2 + z (-1 + z + z^2) + S (1 - 2 z + 2 z^3) = u z^3 + S u z^3$$

From previous expression, write S=... to obtain F

$$s^2 \frac{(-1 + z) z^2 - u z^3 - s u z^3 + z (-1 + z + z^2)}{-1 + 2 z - 2 z^3}$$

$$s - \frac{s^2 (-1 + z) z^2 - u z^3 - s u z^3 + z (-1 + z + z^2)}{-1 + 2 z - 2 z^3}$$

$$s - z - 2 s z + z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^3 - s u z^3$$

$$\frac{1 - 2 z + 2 z^3}{1 - 2 z + 2 z^3}$$

$$s - z - 2 s z + z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^3 - s u z^3$$

$$\frac{1 - 2 z + 2 z^3}{1 - 2 z + 2 z^3}$$

$$2 s \frac{(-1 + z) z^2 - u z^3}{-1 + 2 z - 2 z^3}$$

$$1 - \frac{2 s (-1 + z) z^2 - u z^3}{-1 + 2 z - 2 z^3}$$

$$\frac{1 - 2z - 2sz^2 + 2z^3 + 2sz^3 - uz^3}{1 - 2z + 2z^3}$$

$$1 - 2z - 2sz^2 + 2z^3 + 2sz^3 - uz^3$$

$$1 - 2z + 2z^3$$

Denominator of (s-f) same as that of (1-dfs)

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$z^2 - 5z^3 + 8z^4 - 4z^5 - 2uz^5 + 6uz^6 - 8uz^7 + 4uz^8 + u^2z^8 - u^2z^9$$

$$z[u]^2 - 5z[u]^3 + 8z[u]^4 - 4z[u]^5 - 2uz[u]^5 + 6uz[u]^6 - 8uz[u]^7 + 4uz[u]^8 + u^2z[u]^8 - u^2z[u]^9$$

$$- 2z[u]^5 + 6z[u]^6 - 8z[u]^7 + 4z[u]^8 + 2uz[u]^8 - 2uz[u]^9 + 2z[u] \text{Derivative}[1][z][u] -$$

$$15z[u]^2 \text{Derivative}[1][z][u] + 32z[u]^3 \text{Derivative}[1][z][u] - 20z[u]^4 \text{Derivative}[1][z][u] -$$

$$10uz[u]^4 \text{Derivative}[1][z][u] + 36uz[u]^5 \text{Derivative}[1][z][u] -$$

$$56uz[u]^6 \text{Derivative}[1][z][u] + 32uz[u]^7 \text{Derivative}[1][z][u] +$$

$$8u^2z[u]^7 \text{Derivative}[1][z][u] - 9u^2z[u]^8 \text{Derivative}[1][z][u]$$

$$- 2z[u]^5 + 6z[u]^6 - 8z[u]^7 + 4z[u]^8 + 2uz[u]^8 - 2uz[u]^9 +$$

$$z[u] (2 - 15z[u] + 32z[u]^2 - 10(2+u)z[u]^3 + 36uz[u]^4 - 56uz[u]^5 + 8u(4+u)z[u]^6 - 9u^2z[u]^7)$$

$$\text{Derivative}[1][z][u]$$

$$\{ \{ \text{Derivative}[1][z][u] \rightarrow$$

$$- (2(z[u]^4 - 3z[u]^5 + 4z[u]^6 - 2z[u]^7 - uz[u]^7 + uz[u]^8)) / (-2 + 15z[u] - 32z[u]^2 +$$

$$20z[u]^3 + 10uz[u]^3 - 36uz[u]^4 + 56uz[u]^5 - 32uz[u]^6 - 8u^2z[u]^6 + 9u^2z[u]^7) \} \}$$

$$- 0.0403252$$

According to Drmota's Theorem 1, the mean equals $-z'[1]/z[1]$, computed next.

0.105573

0.0322602

Now compute variance, which by Drmota is $-z''[1]/z[1] + \mu^2 + \mu$

0.0322602

Now compute the standard deviation sigma

0.179611